
FbBalise – Gestion de Balise 

FbBalise est un gestionnaire de balise basé sur un module microcontrôleur ESP32s3. 

 

Il permet de générer des séquences programmées par script et enregistrées en mémoire flash. Une interface WEB 

permet créer ou modifier les scripts et configurer le système. La taille maximale d'un script est 4K caractères. 

L'accès se fait par un ordinateur ou smartphone en se connectant au réseau Wifi « FbBalise » avec le mot de passe 

« 12345678 » par défaut. Le nom du point d'accès et son mot de passe sont modifiables par configuration. 

FbBalise peut également être connecté à un réseau WiFi local. 

Caractéristiques principales 

● Interface de configuration par navigateur (chrome, opéra, firefox) via Wifi 

● Alimentation 7-18V (moins de 40mA à 14v sans Wifi) 

● Modes CW, OPERA, PI4, FT8, JT4, JT9, JT65, Q65, MSK144 et WSPR 

● Synchronisation du temps et calcul QTH locator par GPS 

● Possibilité de piloter un DDS AD9954 

● Dimensions : (cm) 5.25 x 7.15 

  



1. Démarrage 

FbBalise est configuré par des pages WEB accessibles par Wifi. A la mise sous tension, quelques secondes sont 

nécessaires pour que le Wifi soit activé. 

Depuis votre ordinateur personnel ou votre smartphone, cherchez le réseau Wifi « FbBalise » et connectez-le. 

Après la connexion au Wifi « FbBalise », entrez le mot de passe si nécessaire (« 12345678 » par défaut), une 

fenêtre d'accueil s'affiche au bout de quelques secondes. Cette fenêtre va permettre d'accéder par un menu aux 

différentes pages de l'application. 

Si vous êtes bien connecté au Wifi « FbBalise » mais la fenêtre ne s'affiche pas, vous pouvez la connecter à 

l'adresse http://192.168.4.1 

FbBalise utilise le port 80 pour le protocole http et pour le protocole ws. 

Dans le cas où FbBalise est connecté à un réseau local, et si vous voulez avoir un accès public, le port externe 80 

devra être ouvert. Vous pouvez bien sûr rediriger un autre port externe vers le port 80 de FbBalise. 

La LED tricolore donne une information sur le fonctionnement : 

● 2 clignotements rouges : Phase d'initialisation et recherche des réseaux WiFi. 

● Clignotements verts : carte en mode Point d'Accès WiFi 

● Clignotements bleus : carte connectée au réseau WiFi. 

  



2. Page accueil 

La page d'accueil propose un accès à : 

● La configuration par la roue dentée en haut à droite de la fenêtre, 

● Quatre onglets permettant de gérer et exécuter quatre scripts différents. 

○ Si un nom d'onglet est suivi d'un astérisque, celui-ci a été modifié. 

● Une fenêtre texte éditable qui permet de créer ou modifier le script, avec la position du curseur sous la 

fenêtre. 

● Une ligne affichant un éventuel message d'erreur ou une ligne de débug. 

● Un bouton permettant de démarrer, redémarrer ou sauvegarder un script suivant le contexte 

● Une ligne de status donnant l'état d'exécution du script. 

 

Illustration de principe. Le numéro de version affiché en bas de l'écran correspond à la version du logiciel au moment de la capture 

et peut différer de la version décrite dans ce document (0.11.3). 

  



3. Page Paramètres 

Un appui sur la roue dentée en haut à droite permet d'aller sur la page des paramètres. 

Cette page permet de modifier la configuration par défaut et a quatre onglets : 

● Configuration WiFi, 

● Paramètres généraux, 

● SPI 

● mise à jour. 

⚠  Attention 

Toute modification devra être enregistrée par le bouton « Validation » 

 

3 boutons en bas de page vous permettent de choisir un type de redémarrage : 

● « Redémarrer » : le FbBalise est redémarré, les paramètres sont conservés, aucune modification n'est faite 

au PWS. 

● « Effacer la flash » : les scripts contenus dans la mémoire flash sont effacés. Le FbBalise redémarre et la 

configuration est conservée. 

● « Paramètres usine » : tous les paramètres et scripts sont effacés, le FbBalise redémarre en configuration 

par défaut. 

3.1. Configuration WiFi 

 



Paramètres WiFi 

● Nom du réseau WiFi (SSID) pour le point d'accès (AP). 

● Mot de passe pour l'accès à ce réseau WiFi. Par défaut « 12345678 ». 

Option « Délai WiFi » 

● Cette option n'est active qu'en mode point d'accès (AP). Si le FbBalise est connecté à un réseau WiFi, il ne 

sera pas déconnecté. 

● Le Wifi s'active à la mise sous tension de FbBalise. 

● Si l'option est active, le Wifi s'arrêtera automatiquement au bout du temps d'inactivité spécifié en minutes, 

uniquement en mode point d'accès (AP). 

● Le délai minimal est de 5 minutes. Toute action de changement de page ou de sauvegarde redonne le délai 

spécifié. 

● Si l'option est validée, le point d'accès WiFi ne sera de nouveau accessible qu'après coupure et remise sous 

tension ou dévalidation/validation par « activation ». 

Activation 

● Il est possible de désactiver ou activer le point d'accès WiFi à l'aide d'un niveau 0 (désactivé) ou 3.3v 

(activé) sur l'entrée spare spécifiée. Si l'entrée spare n'est pas connectée, le point d'accès est activé par 

défaut. 

Accès par WiFi internet 

● Si un réseau WiFi et son mot de passe sont spécifiés, FbBalise tentera de se connecter à ce réseau. S'il 

échoue, il lancera son propre accès WiFi. 

Accès à la configuration par nom d'utilisateur / mot de passe 

● Si ces champs sont renseignés, un nom d'utilisateur et un mot de passe seront demandés à la connexion. 

  



3.2. Paramètres Généraux 

 

Illustration de principe. Le numéro de version affiché en bas de l'écran correspond à la version du logiciel au moment de la capture 

et peut différer de la version décrite dans ce document (0.11.3). 

Sortie du signal KEY télégraphie 

● Sélection du port de sortie CW Key (spare 1 à 6) 

Inversion signal KEY télégraphie 

● Inverse le signal de sortie CW 

Script 1 

● Permet de sélectionner un spare qui verrouillera la balise sur le script 1. 

Géné externe 

● Spécifie trois spares qui donneront la phase utilisée pour la configuration d'un générateur externe. Voir §9 

Générateur externe. 

Sortie audio 

● Valide ou dévalide la génération du signal audio. Un redémarrage est nécessaire pour prendre en compte la 

modification. 

  



3.3. Configuration SPI 

 

Si les deux switches SPI et AD9954 sont validés, la page de configuration SPI permet de programmer un DDS de 

type AD9954. Cf §12. 

  



3.4. Mise à jour 

 

Illustration de principe. Le numéro de version affiché en bas de l'écran correspond à la version du logiciel au moment de la capture 

et peut différer de la version décrite dans ce document (0.11.3). 

Cette page permet de mettre à jour le logiciel avec une nouvelle version de logiciel. Le fichier est de la forme « 

nom_de_fichier.bin ». 

Après avoir saisi le nom de fichier, appuyer sur le bouton « mettre à jour ». Le fichier sera transmis au module 

balise pour sa mise à jour. En fin de mise à jour, le module sera redémarré. 

⚠  Attention 

Ne pas débrancher le module pendant la phase de mise à jour. 

 

Il est également possible de sauvegarder et restaurer la configuration dans un fichier JSON. 

  



4. Mots clé du script 

● Un paramètre entre crochets est optionnel 

● Une expression peut être 

○ une valeur numérique 

○ une chaine de caractères entre guillemets 

○ une expression entre parenthèses, contenant des expressions, variables, valeurs numériques, 

parenthèses et signes +, -, /, *, % 

○ une condition entre parenthèses, contenant des opérations, variables, valeurs numériques, 

parenthèses et signes >, >=, <, <=, = 

4.1. DEBUG / ECHO 

● DEBUG a un paramètre texte affiché sur la page d'accueil, sous la fenêtre d'édition. 

● ECHO est identique au mot-clé DEBUG mais n'affiche pas l'information DEBUG + N° de ligne. 

● Utiliser des guillemets pour définir l'argument car il peut contenir des espaces. 

● Si l'argument contient une variable, son contenu sera transformé et affiché en texte. 

● Une variable peut être formatée cf §5.3 Formatage d'une variable. 

Dans l'exemple suivant, on suppose un pont diviseur externe d'un rapport de 5.12. La variable sera toujours 

affichée avec une décimale. 

DEBUG "TENSION: $($_A2 * 5.12):1 VOLTS UPTIME : $_UPTIME" 

4.2. DELAY 

● DELAY a un paramètre expression spécifiant le délai de pause en millisecondes 

DELAY (EXPRESSION) 

4.3. HEX 

⚠  Attention 

La fonction HEX décrite ci-dessous n'est pas encore implémentée dans la version 0.11.3. Le 

comportement et la syntaxe indiqués sont ceux prévus pour une prochaine version. 

 

● Fonction qui transforme un nombre ou expression en une chaine hexadécimale. Elle attend deux 

paramètres sous forme d'expression : 

○ Un nombre dont seule la partie entière sera prise en compte 

○ Le nombre de bits du nombre hexadécimal. Si le nombre est plus grand que cette longueur, il sera 

limité à cette longueur. 

HEX((4.52*10), 12) RETOURNE LA CHAINE « 02D » 

● Cette fonction ne réserve pas le signe, c'est de la responsabilité de l'intégration. Donc en 12 bits, 4095 et -1 

retourneront tous deux « FFF » 

4.4. IF / ELSIF / ELSE / ENDIF 

● IF ou ELSIF ont un paramètre entre parenthèses définissant la condition. 

○ Si la condition est vraie, seule la partie IF/ELSIF sera exécutée. Un seul IF/ELSIF sera exécuté, si 

aucun IF ou ELSIF n'est vrai, le ELSE sera exécuté. 

○ Il peut y avoir plusieurs blocs ELSIF. 

○ Si la condition est égale à 0, seule la partie ELSE sera exécutée. 

● Un ENDIF doit obligatoirement spécifier la fin du test conditionnel. 

IF (CONDITION) 



... 

ELSIF (CONDITION) 

... 

ELSE 

... 

ENDIF 

4.5. LOOP / ENDLOOP 

● LOOP peut avoir un paramètre optionnel spécifiant le nombre de boucles 

● Si le nombre de boucles n'est pas spécifié, la boucle sera infinie 

● Un ENDLOOP doit obligatoirement spécifier la fin de boucle. 

LOOP [(EXPRESSION)] 

... 

ENDLOOP 

  



4.6. MODE 

● Mode attend un ou plusieurs paramètres spécifiant les options du mode 

○ CW (vitesse en mots / minute) [tonalité] — tonalité 800Hz par défaut 

○ OPERA PERIOD [tonalité] — PERIOD parmi OPERA05, OPERA1, OPERA2, OPERA4, 

OPERA8, OPERA32, OPERA65, OPERA2H ; tonalité 1500Hz par défaut 

○ PI4 Type [tonalité] — TYPE parmi PI4, PI4-80, PI4-96, PI4-120 ; tonalité de base 800Hz par 

défaut 

○ JT Type [tonalité] — TYPE parmi FT8, JT4A→JT4G, JT65A→JT65C, JT9A→JT9H, Q65-

15A→Q65-15C, Q65-30A→Q65-30D, Q65-60A→Q65-60E, Q65-120A→Q65-120E, Q65-

300A→Q65-300E, WSPR ; tonalité de base 800Hz par défaut pour JT/Q65xx et 1500Hz pour 

WSPR 

○ JT MSK144 [nb_de_trames] — pas de configuration de tonalité ; fréquence centrale de 1500Hz 

avec un décalage de +/- 500Hz ; une trame dure 72ms, une transmission de 30 secondes demande 

environ 400 trames. 

MODE CW (EXPRESSION) [(EXPRESSION)] 

MODE OPERA OPERA05 [(EXPRESSION)] 

MODE PI4 PI4 [(EXPRESSION)] 

MODE JT JT4G [(EXPRESSION)] 

4.7. RESTART 

● Ne nécessite pas de paramètre, 

● Redémarre FbBalise. La configuration et les scripts sont conservés. 

IF ($_UPTIME >= 24) # Redémarre FbBalise toutes les 24 heures 

  RESTART 

ENDIF 

4.8. SEND 

● SEND permet d'envoyer un texte passé en argument en utilisant le mode précédemment défini par le mot-

clé MODE. 

● Utiliser des guillemets pour définir l'argument car il peut contenir des espaces. 

● Si l'argument contient une variable, son contenu sera transformé en texte 

● Une variable peut être formatée cf §5.3 Formatage d'une variable. 

● Le contenu peut être limité par le mode utilisé. 

● Dans le cas de OPERA, seul un indicatif valide sera accepté en argument. 

● Dans le cas du JT, l'indicatif et le qth locator peuvent être envoyés, et les messages type 1, 2 ou 3 pour le 

WSPR. 

SEND "F6FBB JN12BT " 

SEND "TENSION $($_A2 * 4.1):1 " # 1 chiffre après la virgule 

4.9. SET 

● SET permet de définir ou redéfinir la valeur d'une variable. 

● Une variable doit commencer par le caractère $. 

● Une variable est exclusivement numérique. 

● Une variable peut être utilisée dans une expression. 

● SET a deux paramètres : 

○ Un paramètre spécifiant le nom de la variable. Ce nom doit obligatoirement commencer par une 

consonne et être exclusivement alphanumérique. 

○ Un paramètre expression spécifiant la valeur numérique affecté à la variable 

SET $NOMDEVAR ($AUTREVAR / 12.5) 

SET $NOMDEVAR 12.5 



SET $NOMDEVAR ($NOMDEVAR + 1) # incrémente $NOMDEVAR 

4.10. TUNE 

● TUNE a un paramètre expression spécifiant le temps du signal en millisecondes 

TUNE (EXPRESSION) 

4.11. TIMESTAMP 

● TIMESTAMP a un paramètre texte spécifiant le mot de passe utilisé pour le calcul de la clé. Ce mot de 

passe est OBLIGATOIREMENT en majuscules. Il suit la spécification de G4JNT. 

● Le mode en cours doit être CW. 

● Cette commande ne fonctionne que si le GPS est connecté. 

● L'heure utilisée pour le calcul est l'heure GMT fournie par le GPS. 

● Si l'heure n'est pas valide (i.e. absence de GPS) le message « TIME ? » sera transmis. 

TIMESTAMP MOTDEPASSE 

Le résultat du calcul est également accessible en lecture seule via la variable prédéfinie 

$_TIMEST[MOTDEPASSE] (cf. §5.2), qui peut être utilisée dans une instruction SEND ou DEBUG 

en dehors du mot-clé TIMESTAMP. 

  



5. Variables 

5.1. Déclaration et utilisation de variables 

Les variables sont obligatoirement numériques. Elles sont définies ou redéfinies par le mot-clé « SET ». Une 

variable spéciale $(EXPRESSION) n'a pas besoin de déclaration, contient l'expression à évaluer et représente son 

contenu. 

Si la variable est utilisée dans une ligne « SEND », sa valeur sera éventuellement formatée et transformée en texte. 

Le nombre de variables est limité à 10. 

Un nom de variable peut être abrégé, à condition que l'abréviation reste sans ambiguïté par rapport 

aux autres variables utilisées dans le script. Par exemple, $_SEC peut être utilisé à la place de 

$_SECOND, ou $_HUMI à la place de $_HUMID, tant qu'aucune autre variable ne commence par les 

mêmes lettres. Cette possibilité permet d'écrire des scripts plus compacts, ce qui est utile compte tenu 

de la limite de 4K caractères par script. 

5.2. Variables prédéfinies 

Plusieurs variables relatives au système sont prédéfinies. Il s'agit toujours de variables numériques. Leur nom 

commence toujours par un underscore (tiret bas) excepté pour $(expression). Certaines sont en lecture seule (pas 

d'initialisation nécessaire), d'autres en écriture seule. 

Variable Valeur Commentaire R/W 
$_I1 à $_I6 0 ou 1 État du spare en entrée 0=LOW 1=HIGH R 

$_O1 à $_O6 0 ou <> 0 Positionne le spare 0=LOW sinon HIGH W 

$_A1 à $_A3 0 à 3,3 Tension en entrée sur l'analog en volts R 

$_T0 à $_T9 Float ou Texte Télémétrie pour l'application windows W 

$_PTT 0 ou 1 (PTT on) Positionne la sortie PTT W 

$_HOUR 0 à 23 ou -1 Donnera -1 si le GPS n'est pas actif R 

$_MINUTE 0 à 59 ou -1 Donnera -1 si le GPS n'est pas actif R 

$_SECOND 0.000 à 59.999 ou -1 Donnera -1 si le GPS n'est pas actif. Avec ms R 

$_DAY 1 à 31 ou -1 Donnera -1 si le GPS n'est pas actif R 

$_MONTH 1 à 12 ou -1 Donnera -1 si le GPS n'est pas actif R 

$_YEAR yyyy ou -1 Donnera -1 si le GPS n'est pas actif R 

$_UPTIME Nombre d'heures Nombre d'heures depuis le démarrage (flottant) R 

$_UPSCRIPT Nombre d'heures Nombre d'heures depuis le démarrage du script R 

$_LOOP Nombre de boucles Nombre d'itérations dans la boucle active R 

$_POWER Tension Donne la tension d'alimentation du module R 

$_TEMPE Température Ou -1 si le BME280 non connecté R 

$_HUMID Humidité Ou -1 si le BME280 non connecté R 

$_BAROM Pression barom. Ou -1 si le BME280 non connecté R 

$_ALTITUDE Altitude (mètres) Ou -1 si le BME280 non connecté R 

$_BALTI Corr. altitude mètres Correction de pression barométrique. RW 

$_CALLSIGN Indicatif Texte uniquement tel que défini dans params R 

$_LOCATOR Qth Locator Texte uniquement, calculé à partir du GPS R 

$_TIMEST[.] Time stamp 
Texte uniquement. Correspond au calcul effectué par le mot-

clé TIMESTAMP (cf. §4.11). Password entre crochets. 
R 

 

La variable $_TIMEST[PASSWORD] contient le timestamp au format spécifié par G4JNT, i.e. la 

minute en 2 chiffres, un espace et 3 lettres. Le password entre crochets est OBLIGATOIREMENT en 

caractères majuscules. 

 

La variable $_BALTI a la valeur 0 par défaut, donc pas de correction de la pression. La valeur devra 

être initialisée avec l'altitude pour la correction. 

 



⚠  Attention 

Les variables $_TEMPE, $_HUMID et $_BAROM nécessitent la présence d'un BME280 connecté 

sur le port I2C. La détection d'adresse est automatique. 

 

⚠  Attention 

Les variables $_I1 à $_I6 et $_A1 à $_A3 positionnent le port spare en entrée. Les variables $_O1 à 

$_O6 positionnent le port spare en sortie. Si le port spare concerné est utilisé pour un autre usage, cela 

peut affecter son fonctionnement. 

 

⚠  Attention 

La tension en entrée sur un port spare ne doit pas dépasser 3,3v sous peine de détruire le 

microcontrôleur. Les mesures de tension supérieures peuvent se faire en utilisant un pont diviseur 

externe. La valeur peut alors être multipliée en utilisant une expression. 

 

Exemples : 

SET $_O2 1   # Positionne le port spare 2 au niveau haut 

SET $_O2 0   # Positionne le port spare 2 au niveau bas 

IF ($_I1)    # Vrai si le port spare 1 est à niveau haut 

IF ($_A3 > 2.1)   # Vrai si le port analog 3 est à une tension supérieure à 2,1v 

SET $_T0 "TENSION BATTERIE: $_POWER:2,V"   # Envoie la tension d'alimentation vers 
l'application Windows sur la télémétrie 0 

  



5.3. Formatage d'une variable 

Lors de l'utilisation d'une variable sous forme textuelle (dans DEBUG ou SEND par exemple), il est possible de 

formater la variable en utilisant le format suivant : 

$VAR:N,S 

● N est le nombre de décimales. La valeur sera arrondie en fonction de ce nombre. 

● S est le caractère de séparation qui remplacera le point décimal. 

● Le formatage est optionnel, mais devra être utilisé dans cet ordre 

Exemples : 

SET $PI 3.1415926536   # Initialise la variable avec la valeur 3.1415926536 

SEND "$PI:3,V"   # Enverra le texte 3V142 

SEND "$PI:,V"    # Enverra le texte 3V14159 

Variables tension d'alimentation et météo pour PI4 : 

SEND "/_PS$_POWER:1,V"   # Envoie la tension 12.4v : '/_PS12V4' 

SEND "/_HUMI$_HUMI:0"    # Envoie l'hygrométrie 56% : '/_HUMI56' 

IF ($_TEMPE >= 0)   # Cas pour +26° et -12° 

  SEND "/_TMPO$_TEMPE:0"   # Envoie la température 26° : '/_TMPO26' 

ELSE 

  SEND "/_TMNE$(-$_TEMPE):0"   # Envoie la température -12° : '/_TMNE12' 

ENDIF 

SET $_BALTI 400   # Correction de la pression barom avec l'altitude (400m) 

SEND "/_BARO$($_BAROM%100):0"   # Envoie la pression 1016hPa : '/_BARO16' 

  



6. Expressions 

Une expression est de la forme « A opérateur B » où : 

● A est une valeur, une variable ou une expression. 

● Un opérateur du tableau ci-dessous, 

● B est une valeur, une variable ou une expression. 

Les opérations sont réalisées en virgule flottante. Les opérateurs sont : 

Opérateur Action Commentaire 

+ Addition  

- Soustraction  

* Multiplication  

/ Division Quotient de la division flottante 

% Modulo Reste de la division flottante 

 

Les multiplications, divisions et modulo sont effectuées en priorité, de gauche à droite. Les additions et 

soustractions sont effectuées en deuxième priorité, également de gauche à droite. 

Une expression peut être entourée de parenthèses, auquel cas les expressions entre parenthèses sont effectuées en 

priorité. 

Exemples : 

(3 + 5 * 2)   => 3 + 10 => 13 

((3 + 5) * 2) => 8 * 2  => 16 

(3 * 5 + 2)   => 15 + 2 => 17 

  



7. Comparaisons et opérateurs logiques 

Une comparaison est de la forme « A Comparateur B » où : 

● A est une valeur, une variable ou une expression. 

● B est une valeur, une variable ou une expression. 

Les calculs sont faits en virgule flottante. L'égalité peut donc être difficile à trouver. Le résultat d'une comparaison 

est égal à 1.0 si le résultat est vrai sinon 0.0 si le résultat est faux. 

Les opérateurs logiques fonctionnent également sur des valeurs égales à 0 (0 logique) ou différentes de 0 (1 

logique). Les opérateurs sont : 

Comparateur Action Commentaire 

< Inférieur  

> Supérieur  

<= Inférieur ou égal  

>= Supérieur ou égal  

!= ou <> Différent  

== ou = Égalité  

& Et logique  

| Ou logique  

^ Ou exclusif logique  

 

Exemples : 

IF ($_A3 > 2.1)   # Vrai si le port analog 3 est à une tension supérieure à 2,1v 

  Exécute si analog3 supérieur à 2.1v 

ELSE 

  Exécute si le port analog 3 est à une tension inférieure ou égale à 2,1v 

ENDIF 

IF ($_T & 2.1)   # Vrai si la valeur de $_T est différente de 0 

  



8. Réalisation du script 

8.1. Création du script 

Le script est entré au clavier (ou collé) dans la fenêtre d'édition. Lorsque l'édition est terminée, le bouton de 

sauvegarde va envoyer et sauvegarder le script dans la flash de l'ESP32. 

● Le script sera interprété en majuscules et ne fait pas de différence entre majuscules et minuscules. 

● Toute ligne devra commencer par un mot clé, suivi de ses arguments éventuels. 

● Toute ligne vide est ignorée. 

● Toute fin de ligne sera ignorée après le caractère # 

8.2. Script simple 

Ce script va permettre de : 

● Sélectionner le mode CW à 15 mots/minute, son à 800Hz 

● Envoyer l'indicatif et le qth locator suivi d'une pause de 1s, tune de 5s, pause de 1s 

● Le tout en boucle infinie. 

MODE CW 15 800 

LOOP 

  SEND "F6FBB JN12BT " 

  DELAY 1000 

  TUNE 5000 

  DELAY 1000 

ENDLOOP 

8.3. Script plus évolué 

Ce script va générer : 

● Un mode CW 15 mots minute à 800Hz, 

● trois fois l'indicatif, le qth locator et une pause de 500ms, 

● ensuite, pause de 500ms puis tune de 5 secondes puis pause de 1 seconde 

● il bouclera indéfiniment… 

MODE CW 15 800 

LOOP 

  LOOP 3 

    SEND "F6FBB JN12BT " 

    DELAY 500 

  ENDLOOP 

  DELAY 500 

  TUNE 5000 

  DELAY 1000 

ENDLOOP 

  



8.4. Script avec test conditionnel 

Ce script va générer : 

● Un mode CW 15 mots minute à 800Hz, 

● trois fois l'indicatif, une pause de 1s puis deux fois le qth locator, 

● ensuite, pause de 500ms puis tune de 5 secondes puis pause de 1 seconde 

● Toutes les 4 boucles il enverra l'indicatif en mode OPERA puis retour en mode CW 

● il bouclera indéfiniment… 

MODE CW 15 800 

LOOP 

  LOOP 3 

    SEND "F6FBB " 

  ENDLOOP 

  DELAY 1000 

  LOOP 2 

    SEND "JN12BT " 

  ENDLOOP 

  DELAY 1000 

  TUNE 5000 

  IF (($_LOOP % 4) = 0) 

    DELAY 1000 

    MODE OPERA OPERA05 1500 

    SEND "F6FBB" 

    MODE CW 15 800 

  ENDIF 

  DELAY 1000 

ENDLOOP 

8.5. Script avec synchronisation à la minute 

Ce script va permettre d'envoyer 3 fois l'indicatif suivis d'une pause d'une seconde, puis un tune jusqu'à la seconde 

59, faire une pause de 1 seconde puis d'envoyer l'indicatif en mode OPERA. Il nécessite la présence du module 

GPS pour avoir l'information horaire. 

MODE CW 15 800 

LOOP 

  LOOP 3 

    SEND "F6FBB " 

  ENDLOOP 

  DELAY 1000 

  TUNE ((59 - $_SEC) * 1000) 

  DELAY 1000 

  MODE OPERA OPERA05 1500 

  SEND "F6FBB" 

  DELAY 1000 

  MODE CW 15 800 

ENDLOOP 

  



9. Générateur externe 

Dans l'onglet « Fréquences » du menu configuration, il est possible de sélectionner 3 spares pour activer un 

générateur externe OOK ou de fréquences. 

Le niveau des spares attribués à F1, F2 et F3 suivra la logique suivante : 

F3 F2 F1 Fréquence à générer 

0 0 0 CW_OFF / SPACE — Génération OOK ou shift suivant hardware 

0 0 1 CW_ON / MARK — Génération OOK ou shift suivant hardware 

0 1 0 OPERA_OFF / SPACE — Génération OOK ou shift suivant hardware 

0 1 1 OPERA_ON / MARK — Génération OOK ou shift suivant hardware 

1 0 0 PI4/0 — PI4 fréquence 0 

1 0 1 PI4/1 — PI4 fréquence 1 

1 1 0 PI4/2 — PI4 fréquence 2 

1 1 1 PI4/3 — PI4 fréquence 3 

 

● La sélection des 3 spares n'est pas obligatoire. Il est possible de sélectionner tout ou partie de ces spares. 

○ F1 => « KeyUp / KeyDown » dans le cas d'une balise CW uniquement 

○ F2 => CW / OPERA si PI4 n'est pas utilisé 

○ F3 => mode PI4 

 

Exemple : ce script active le générateur externe en mode CW (F2=0), envoie l'indicatif, puis positionne F1 

(KeyUp/KeyDown) en fonction du texte transmis, avant de repasser les spares au niveau bas : 

SET $_O2 0   # F2 = 0 : mode CW 

MODE CW 15 800 

LOOP 

  SET $_O1 1   # F1 = 1 : CW_ON / MARK 

  DELAY 500 

  SET $_O1 0   # F1 = 0 : CW_OFF / SPACE 

  DELAY 500 

ENDLOOP 

  



10. Schéma électronique 

 

  



11. Connecteurs et jumpers 

 

● SJ1 : mesure de la tension d'alimentation sur AN1 

● SJ2 : sortie OOK open drain 

● SJ3 : sortie PTT open drain 

● SJ4 : sortie OOK SPI 

 

● J1 : antenne GPS 

● J3 : Antenne WiFi 

  



12. Interface et carte DDS AD9954 

Une carte basée sur le circuit DDS AD9954 de Analog Devices peut être directement pilotée via le bus SPI. 

Si le switch SPI et le switch AD9954 sont activés, une fenêtre spécifique permet de configurer le circuit et de 

générer le signal HF en FSK ou OOK suivant les modes et configurations. 

 

Carte fille DDS AD9954 

 

Mesure au spectrum analyzer 



 

La valeur de la fréquence de base doit être renseignée. La fréquence maximale est de 160 000 000 Hz. 

Toutes les valeurs de shift et de fréquences sont exprimées en Hz. 

Le multiplicateur permet d'adapter les valeurs de décalage des modes de modulation multifréquences en fonction 

d'une multiplication de la fréquence générée. I.E. dans le cas d'une génération 144MHz multipliée par 3 pour 

générer du 432MHz, le décalage de 315Hz en JT4G sera ramené à 105Hz, de façon à retrouver la valeur nominale 

après les étages de multiplication de l'émetteur. Les autres modes PI4 et WSPR auront le même fonctionnement. 

  



 

Génération de la modulation OOK avec fronts « anti-claquements » 

 

Câble FbBalise vers connecteur DDS 

  



13. Aide-mémoire des mots-clés 

Ce tableau récapitule, à titre de référence rapide, l'ensemble des mots-clés du langage de script décrits au §4. 

Mot-clé Rôle 
DEBUG / ECHO Affiche un texte (avec variables) sur la page d'accueil 

DELAY (EXPRESSION) Pause en millisecondes 

HEX(n, bits) Convertit un nombre en chaîne hexadécimale (non implémenté en 0.11.3) 

IF / ELSIF / ELSE / 
ENDIF 

Test conditionnel 

LOOP [(n)] / ENDLOOP Boucle, infinie si n absent 

MODE ... Sélectionne le mode de transmission (CW, OPERA, PI4, JT, MSK144…) 

RESTART Redémarre FbBalise (config. et scripts conservés) 

SEND "texte" Émet un texte dans le mode courant 

SET $VAR (EXPRESSION) Définit ou modifie une variable 

TUNE (EXPRESSION) Émet une porteuse pendant un temps donné (ms) 

TIMESTAMP MOTDEPASSE Émet un timestamp chiffré (norme G4JNT), nécessite le GPS et le mode CW 

  



14. Remerciements 

La réalisation logicielle est de ma conception. Ce logiciel fonctionne sur une carte FbBalise et pourra utiliser une 

carte fille GPS (pour la mise à l'heure et la localisation) 

Le logiciel est écrit en C/C++ sous l'environnement « platformio / vscode » et inclut plusieurs librairies tierces 

ESP32. 

Merci à F5DJL et F1HDI pour leurs idées, remarques et soutien ! 

Jean-Paul ROUBELAT, F6FBB. 


